
Lecture Outline
XPath and XQuery

• Query expressions
Direct and computed constructors
FLWOR expressions
CondiƟonal expressions
QuanƟfied expressions

XQuery
XML Query Language



Sample Data

<?xml version="1.1" encoding="UTF-8"?>
<movies>

<movie year="2006" rating="76" director="Jan Svěrák">
<title>Vratné lahve</title>
<actor>Zdeněk Svěrák</actor>
<actor>Jiří Macháček</actor>

</movie>
<movie year="2000" rating="84">

<title>Samotáři</title>
<actor>Jitka Schneiderová</actor>
<actor>Ivan Trojan</actor>
<actor>Jiří Macháček</actor>

</movie>
<movie year="2007" rating="53" director="Jan Hřebejk">

<title>Medvídek</title>
<actor>Jiří Macháček</actor>
<actor>Ivan Trojan</actor>

</movie>
</movies>

Expressions
XQuery expressions

• Path expressions (tradiƟonal XPath)
SelecƟon of nodes of an XML tree

• FLWOR expressions
for … let … where … order by … return …

• CondiƟonal expressions
if … then … else …

• QuanƟfied expressions
some|every … satisfies …

Expressions
XQuery expressions

• Boolean expressions
and, or, not logical connecƟves

• Primary expressions
Literals, variable references, funcƟon calls, constructors, …

• …

path expressionpath expression

FLWOR expressionFLWOR expression

conditional expressionconditional expression

switch expressionswitch expression

quantified expressionquantified expression

boolean expressionboolean expression

primary expressionprimary expression

......

Node Constructors
Constructors

• Allow for creaƟon of new nodes for elements, aƩributes, …
I.e. nodes that do not exist in the original XML document

Direct constructor
• Well-formed XML fragment with embedded query expressions

E.g.: <movies>{ count(//movie) }</movies>
Computed constructor

• Special syntax
E.g.: element movies { count(//movie) }

Node Constructors
Direct constructor

• The enƟre expression must be a well-formed XML fragment
Names of elements and aƩributesmust be fixed

<< namename

attribute constructorattribute constructor

// >>

<< namename

attribute constructorattribute constructor

>>

element content constructorelement content constructor << // namename >>

• Embedded query expressions can be used
However, only in aƩribute values and element content!

Node Constructors
Direct constructor

• AƩribute
namename == "" characterscharacters

{{ expressionexpression }}

""

• Element content
characterscharacters

direct constructordirect constructor

{{ expressionexpression }}

• Embedded query expressions
Enclosed by curly braces {}

– Escaping sequence: {{ and }}

Node Constructors: Example
Create a summary of all movies
<movies>

<count>{ count(//movie) }</count>
{

for $m in //movie
return

<movie year="{ data($m/@year) }">{ $m/title/text() }</movie>
}

</movies>

<movies>
<count>3</count>
<movie year="2006">Vratné lahve</movie>
<movie year="2000">Samotáři</movie>
<movie year="2007">Medvídek</movie>

</movies>

Node Constructors
Computed constructor

• Names of elements and aƩributes can be dynamic

• Element node
elementelement element nameelement name

{{ expressionexpression }}

{{ expressionexpression

,,

}}

• AƩribute node
attributeattribute attribute nameattribute name

{{ expressionexpression }}

{{ expressionexpression }}

• Text node
texttext {{ expressionexpression }}

Node Constructors: Example
Create a summary of all movies
element movies {

element count { count(//movie) },
for $m in //movie
return

element movie {
attribute year { data($m/@year) },
text { $m/title/text() }

}
}

<movies>
<count>3</count>
<movie year="2006">Vratné lahve</movie>
<movie year="2000">Samotáři</movie>
<movie year="2007">Medvídek</movie>

</movies>

FLWOR Expressions
FLWOR expression (XQuery 1.0)

• Allow for advanced iteraƟons over sequences of items
for clausefor clause

let clauselet clause

where clausewhere clause order by clauseorder by clause return clausereturn clause

Clauses
• for – selecƟon of items to iterate over
• let – bindings of auxiliary variables
• where – condiƟons to be saƟsfied
• order by – order in which the items are processed
• return – result to be constructed

FLWOR Expressions: Example
Find Ɵtles of movies with raƟng 75 and more
for $m in //movie
let $r := $m/@rating
where $r >= 75
order by $m/@year
return $m/title/text()

Samotáři
Vratné lahve

FLWOR Expressions: Clauses
For clause

• Iterates over items of one or more input sequences
These items are accessible via the introduced variables

forfor $$ variable namevariable name

atat $$ variable namevariable name

inin expressionexpression

,,

• OpƟonal posiƟonal variable
Allows to access the ordinal number of the current item

• WhenmulƟple input sequences are provided…
Then the behavior is idenƟcal to the usage of mulƟple
consecuƟve single-variable for clauses

– I.e., as if the for loops are embedded into each other

FLWOR Expressions: Clauses
Let clause

• Defines one or more auxiliary variable assignments
letlet $$ variable namevariable name :=:= expressionexpression

,,

FLWOR Expressions: Clauses
Where clause

• Allows to describe complex filtering condiƟons
• Items not saƟsfying the condiƟons are skipped

wherewhere expressionexpression

Order by clause
• Defines the order in which the items are processed

order byorder by expressionexpression

ascendingascending

descendingdescending

,,

FLWOR Clauses
Return clause

• Defines how the result sequence is constructed
• Evaluated once for each suitable item

returnreturn expressionexpression

Various supported use cases
• Querying, joining, grouping, aggregaƟon, integraƟon,

transformaƟon, validaƟon, …

FLWOR Examples
Find Ɵtles of movies filmed in 2000 or later such that they have at
most 3 actors and a raƟng above the overall average
let $r := avg(//movie/@rating)
for $m in //movie[@rating >= $r]
let $a := count($m/actor)
where ($a <= 3) and ($m/@year >= 2000)
order by $a ascending, $m/title descending
return $m/title

<title>Vratné lahve</title>
<title>Samotáři</title>

FLWOR Examples
Find movies in which each individual actor stared
for $a in distinct-values(//actor)
return <actor name="{ $a }">

{
for $m in //movie[actor[text() = $a]]
return <movie>{ $m/title/text() }</movie>

}
</actor>

<actor name="Zdeněk Svěrák">
<movie>Vratné lahve</movie>

</actor>
<actor name="Jiří Macháček">

<movie>Vratné lahve</movie>
<movie>Samotáři</movie>
<movie>Medvídek</movie>

</actor>
...

FLWOR Examples
Construct an HTML table with data about movies
<table>

<tr><th>Title</th><th>Year</th><th>Actors</th></tr>
{

for $m in //movie
return

<tr>
<td>{ $m/title/text() }</td>
<td>{ data($m/@year) }</td>
<td>{ count($m/actor) }</td>

</tr>
}

</table>

FLWOR Examples
Construct an HTML table with data about movies
<table>

<tr><th>Title</th><th>Year</th><th>Actors</th></tr>
<tr><td>Vratné lahve</td><td>2006</td><td>2</td></tr>
<tr><td>Samotáři</td><td>2000</td><td>3</td></tr>
<tr><td>Medvídek</td><td>2007</td><td>2</td></tr>

</table>

CondiƟonal Expressions
CondiƟonal expression

ifif (( expressionexpression )) thenthen expressionexpression elseelse expressionexpression

• Note that the else branch is compulsory
Empty sequence () can be returned if needed

Example
if (count(//movie) > 0)
then <movies>{ string-join(//movie/title, ", ") }</movies>
else ()

<movies>Vratné lahve, Samotáři, Medvídek</movies>

Switch Expressions
Switch

• The first matching branch is chosen,
its return clause is evaluated
and the result returned

switchswitch (( expressionexpression ))

casecase expressionexpression returnreturn expressionexpression

defaultdefault returnreturn expressionexpression

• The default branch is compulsory
and must be provided as the last opƟon

Switch Expressions
Example

Return movies with aggregated informaƟon about their actors
xquery version "3.0";
for $m in //movie
return

<movie>
{ $m/title }
{

switch (count($m/actor))
case 0 return <no-actors/>
case 1 return <actor>{ $m/actor/text() }</actor>
default return <actors>{ string-join($m/actor, ", ") }</actors>

}
</movie>

QuanƟfied Expressions
QuanƟfier

• Returns true if and only if…
in case of some at least one item
in case of every all the items

• … of a given sequence/s saƟsfy the provided condiƟon
somesome

everyevery

$$ variable namevariable name inin expressionexpression

,,

satisfiessatisfies expressionexpression

QuanƟfied Expressions
Examples

Find Ɵtles of movies in which Ivan Trojan played
for $m in //movie
where

some $a in $m/actor satisfies $a = "Ivan Trojan"
return $m/title/text()

Samotáři
Medvídek

Find names of actors who played in all movies
for $a in distinct-values(//actor)
where

every $m in //movie satisfies $m/actor[text() = $a]
return $a

Jiří Macháček

FLWOR Expressions
Extended FLWOR expression (XQuery 3.0)

for clausefor clause

let clauselet clause

window clausewindow clause

for clausefor clause

let clauselet clause

window clausewindow clause

where clausewhere clause

group by clausegroup by clause

order by clauseorder by clause

count clausecount clause

return clausereturn clause

Clauses
• window – sliding or tumbling windows to iterate over
• group by – equality-based groupings of input items
• count – posiƟonal numbers of tuples in a stream

FLWOR For Clauses
For clause

• OpƟonal allowing empty
One () item is considered instead of an empty sequence
Suitable for outer joins

– Does not eliminate one item when the other would be missing
• PosiƟonal variable

Allows to access the ordinal number of the current item

forfor $$ variable namevariable name

allowing emptyallowing empty atat $$ variable namevariable name

inin expressionexpression

,,

FLWOR Group By Clauses
Group by clause

• Performs equality-based grouping defined by one or more
grouping variables

Only singleton values are permiƩed for these variables
– Otherwise a runƟme error is raised

Each input item will appear only in one output group
• Non-grouping variable is rebound to a sequence of all the

matching items from a given group

group bygroup by $$ variable namevariable name

:=:= expressionexpression

,,

FLWORWindow Clauses
Window clause

• Allows to iterate over the generated windows
Two modes: tumbling and sliding

• Window = sequence of consecuƟve items from the input
Accessible via the main variable
Contains the start item, end item, and all items between them

forfor tumbling windowtumbling window $$ variable namevariable name inin expressionexpression

start conditionstart condition end conditionend condition

sliding windowsliding window $$ variable namevariable name inin expressionexpression

start conditionstart condition end conditionend condition

FLWORWindow Clauses
Window start condiƟon

• Start item is an item that saƟsfies a given condiƟon

startstart window variableswindow variables whenwhen expressionexpression

Window end condiƟon
• End item is the first item (beginning with the start item) that

saƟsfies a given condiƟon
• When such an item cannot be found…

Then the last item is the very last input item
But only in case the only keyword is not specified
Otherwise such a window is not generated at all

onlyonly

endend window variableswindow variables whenwhen expressionexpression

FLWORWindow Clauses
Window variables (all of them are opƟonal)

• Bound to the first/last item
• at: bound to the ordinal posiƟon of the first/last item
• previous: bound to the item that precedes the first/last item
• next: bound to the item that follows the first/last item

$$ variable namevariable name

atat $$ variable namevariable name

previousprevious $$ variable namevariable name nextnext $$ variable namevariable name

NDBI001: Query Languages I | Lecture 8: XML Databases: XQuery | 8. 12. 2020 33



FLWORWindow Clauses
Tumbling window

• Search for the start item of the next window begins with the
item that follows the end item of the previous window (or at
the very beginning)

• ⇒ windows never overlap
Input item may never be found in mulƟple windows

• When the end condiƟon is missing…
All start items are first detected
Each window is terminated by the item that precedes the next
starƟng one (or by the last input item at the very end)

FLWORWindow Clauses
Sliding window

• Every item that saƟsfies the start condiƟon becomes the
starƟng item of a new window

• ⇒ windows may overlap
Input item may be found in mulƟple windows

FLWOR Count Clauses
Count clause

• Allows to access the ordinal number of the current tuple in a
stream

countcount $$ variable namevariable name

Final ObservaƟons
XQuery

• Keywordsmust always be in lowercase
• XQuery is a funcƟonal query language
• Whenever expression is menƟoned in any diagram,

expression of any kind can be used (without any limitaƟons)

Lecture Conclusion
XPath expressions

• Absolute and relaƟve paths
• Axes, node tests, and predicates

XQuery expressions
• Constructors: direct, computed
• FLWOR expressions
• CondiƟonal, quanƟfied, comparison, …

